I recently received a request from a reader to outline the use of AutoCAD® selection set filters when coding in VBA. The AutoCAD selection set object is not the most intuitive of objects to master. It's fussy, full of frustrating hurdles, and if you're trying to use the object's internal filtering capability, you must have a handle on entity group codes to make any sense of it. 
Before we examine the internal filtering capability of a selection set object, let's look at how to: 
· Create and manipulate a selection set. 
· Achieve basic selection set filtering by examining and testing for basic entity properties. 
· Easily determine an entity's group codes for more sophisticated selection set filtering. 
The Selection Set Object
If you're a typical AutoCAD user, you use selection sets to manipulate one or more drawing entities at a time using editing commands (MOVE, ROTATE, SCALE, and so on). To work with a selection set in VBA, three things have to happen: 
1. You must dimension a selection set object. 

2. You must add that object to the collection of existing selections set objects for the drawing. 

3. You must "fill" that object with the entities to be treated. 
Look at the following routine, which demonstrates these three requirements by selecting entities and deleting them using a selection set's SelectOnScreen mode. 
Private Sub CommandButton1_Click()
   Dim ssetObj As AcadSelectionSet
   UserForm1.Hide
   Set ssetObj = ThisDrawing.SelectionSets.Add("SS01")
   ssetObj.SelectOnScreen
   ssetObj.Erase
   ssetObj.Delete
   UserForm1.Show
End Sub 
What is happening here? Let's go through the seven lines of indented code in this routine, assigned to a CommandButton when it is clicked: 
1. A selection set object is dimensioned as an AcadSelectionSet object type. 

2. The UserForm is hidden so that you can use the pickbox to select objects in the graphics area. 

3. The collection of existing selection set items is augmented with a new selection set item named "SS01," and the newly dimensioned selection set object is set to this item in the collection. 

4. The user now selects entities on the screen using standard selection modes (fence, crossing, window, and so on). 

5. When completed, the entities collected in the selection set object are erased. The selection set object, empty though it is, still exists. 

6. The select set object itself is deleted. This is important because if you don't delete the selection set, the next time you run the routine you will get an error (the selection set item will already exist). 

7. Finally, the UserForm is made visible again. 
Dealing with Bomb-Outs
Bomb-outs are common during development and debugging of any routine, but particularly one that uses a selection set. Why? Typically because you forget to delete the selection set item after the bomb-out. Otherwise, the next time you run the routine, it generates an error because it's trying to re-create an existing selection set. For this reason, many programmers name their selection set objects consecutively. For example, if they start with the name "SS01" they name the next set "SS02" and the one after that "SS03." This naming convention keeps the routine working until it is completely finished and debugged. Try it out. 
What Is Selection Set Filtering?
Suppose that you have a drawing with lines and circles mixed together in a certain area and your goal is to delete only the lines in that area. You could use the above routine to carefully pick only the lines you want to delete. But it would be far lazier (and more productive) if you could simply select both lines and circles in that area and have all the circles automatically removed from the selection set before the selection set was deleted. Let's examine a modified version of the above routine with which you can select both lines and circles but delete only the lines. 
Private Sub CommandButton1_Click()
   Dim ssetObj As AcadSelectionSet
   Dim entObj As AcadEntity
   Dim i As Integer
   UserForm1.Hide
   Set ssetObj = ThisDrawing.SelectionSets.Add("SS01")
   ssetObj.SelectOnScreen
   For i = 0 To ssetObj.Count - 1
      Set entObj = ssetObj.Item(i)
      If entObj.ObjectName = "AcDbLine" Then entObj.Delete
   Next i
   ssetObj.Delete
   UserForm1.Show
End Sub 
When you run this routine, a selection set object is dimensioned, added to the selection set collection, and filled, in this case, by picking circles and lines on the screen. But this time, all items added to the selection set are first examined by their individual ObjectNames, and only those that meet the testing criteria are deleted. 
This is a simple "brute force" way to "filter" selection sets. You could easily replace the ObjectName property in this routine by other entity properties such as Color, Layer, or Linetype and filter the selection set accordingly. 
How would you use this method to filter a selection set using several criteria? Suppose you want to filter a selection set (filled with many different types of entities) for a specific entity type (for our purposes, LINEs) that are on a specific layer ("WALL") and have a specific color (ByLayer)? Just modify the above routine as follows: 
Private Sub CommandButton1_Click()
   Dim ssetObj As AcadSelectionSet
   Dim entObj As AcadEntity
   Dim i As Integer
   UserForm1.Hide
   Set ssetObj = ThisDrawing.SelectionSets.Add("SS04")
   ssetObj.SelectOnScreen
   For i = 0 To ssetObj.Count - 1
      Set entObj = ssetObj.Item(i)
      If entObj.ObjectName = "AcDbLine" Then
      If entObj.Layer = "WALL" Then
      If entObj.Color = acByLayer Then entObj.Delete
      End If
      End If
      End If
   Next i
   ssetObj.Delete
   UserForm1.Show
End Sub
Using Group Codes
The unique properties of all AutoCAD entities are organized by Group Codes. Group 0, for example, usually states the entity type of an entity; Group 5 usually states the handle of an entity; Group 8 usually states the layer of an entity, and so forth. You can find most of the entity Group Codes in the online DXF Reference But there is an incredibly easy way to report an entity's group codes and their states directly from the command prompt. Simply enter the AutoLISP® expression 
(entget (car (entsel))) 
at the command prompt and select an entity on the screen, and the entity's Group Codes and their states are reported to the Text window. Choose a line in any drawing, enter the AutoLISP expression at the command prompt, and select that line. You'll get the following kind of information, reported in the Text window: 
Command: (entget (car (entsel)))
Select object: ((-1 . <Entity name: 40065150>) (0 . "LINE") (330 . <Entity name: 40044cf8>) (5 . "BA") (100 . "AcDbEntity") (67 . 0) (410 . "Model") (8 . "0") (100 . "AcDbLine") (10 0.26899 49.9588 0.0) (11 2.81767 44.3771 0.0) (210 0.0 0.0 1.0)) 
Organized a little more clearly, this information reads: 
(-1 . <Entity name: 40065150>)
(0 . "LINE")
(330 . <Entity name: 40044cf8>)
(5 . "BA")
(100 . "AcDbEntity")
(67 . 0)
(410 . "Model")
(8 . "0")
(100 . "AcDbLine")
(10 0.26899 49.9588 0.0)
(11 2.81767 44.3771 0.0)
(210 0.0 0.0 1.0)) 
What can you interpret from this report? 
For starters, according to Group 0 (which delineates Entity Tyle), the entity type is a "LINE,", and according to Group 5 (which delineates Entity Handle) its handle is "BA." refering to other Group Codes, we can also know that the entity is created in model space instead of a layout tab, and it's on layer "0," its beginning point (a set of 3 real number coordinates, which for this example) is (0.27, 49.95, 0.00) and its end point is (2.82, 44.38, 0.00). You can access the Group Code states for any entity type using this AutoLISP statement. 
The selection set object has optional filtering parameters for several of its modes. If you can provide the Group Code(s) and value(s) you wish to filter on, you can develop a routine that automatically filters a selection set based on that information. 
Look at the following routine, which is a modified version of the first routine we used to select and delete entities. Notice that two arrays have been dimensioned to contain a Group Code integer (grpCode) and a value for the Group Code (dataVal). Specifically, the routine uses Group Code 0 (which delineates Entity Type) and a data value of "LINE" to filter out (from the selection set) all picked entities that are not lines: 
Private Sub CommandButton1_Click()
   Dim entObj As AcadEntity
   Dim ssetObj As AcadSelectionSet
   Dim grpCode(0) As Integer
   Dim dataVal(0) As Variant
   UserForm1.Hide
   grpCode(0) = 0
   dataVal(0) = "LINE"
   Set ssetObj = ThisDrawing.SelectionSets.Add("SS01")
   ssetObj.SelectOnScreen grpCode, dataVal
   ssetObj.Erase
   ssetObj.Delete
   UserForm1.Show
End Sub
This mode even gives you a visual report of the filtering as it is performed, so that only those entities meeting the filtering parameter criteria are highlighted. If you only wanted LINEs (Group Code 0="LINE") on Layer "WALL" (Group Code 8 = "WALL") selected, you could expand the arrays to two items as shown below: 
Private Sub CommandButton1_Click()
   Dim entObj As AcadEntity
   Dim ssetObj As AcadSelectionSet
   Dim grpCode(1) As Integer
   Dim dataVal(1) As Variant
   UserForm1.Hide
   grpCode(0) = 0
   dataVal(0) = "LINE"
   grpCode(1) = 8
   dataVal(1) = "WALL"
   Set ssetObj = ThisDrawing.SelectionSets.Add("SS01")
   ssetObj.SelectOnScreen grpCode, dataVal
   ssetObj.Erase
   ssetObj.Delete
   UserForm1.Show
End Sub 
The process for filtering a selection set for multiple criteria then becomes very easy: simply expand these two arrays (grpCode and dataVal) to accommodate the number of Group Codes being examined and their corresponding values. Because the variable dataVal has been dimensioned as a Variant, it can contain integer values, strings, and other data types. 
Working With Multiple Group Codes
Let's look at a more complicated filtering application. Suppose you have a huge collection of lightweight polylines mixed in with other entity types (text, lines, circles, arcs, and so on), and you want to delete only the "closed" polylines. To filter the drawing properly, you can turn to the online DXF Reference to find out what the Group Code is for detecting the "open" and "closed" states of a lightweight polyline, or you can use the AutoLISP expression defined earlier and apply it to a lightweight polyline (in an "open" and "closed" state) to report the Group Codes you need. Selecting an "open" lightweight polyline with four vertices might report: 
Command: (entget (car (entsel))) 
Select object: ((-1 . <Entity name: 40044e68>) (0 . "LWPOLYLINE") (330 . <Entity name: 40044cf8>) (5 . "65") (100 . "AcDbEntity") (67 . 0) (410 . "Model") (8 . "1") (100 . "AcDbPolyline") (90 . 4) (70 . 0) (43 . 0.0) (38 . 0.0) (39 . 0.0) (10 7.86986 10.6993) (40 . 0.0) (41 . 0.0) (42 . 0.0) (10 9.54074 7.86177) (40 . 0.0) (41 . 0.0) (42 . 0.0) (10 11.1782 10.232) (40 . 0.0) (41 . 0.0) (42 . 0.0) (10 9.57416 12.0347) (40 . 0.0) (41 . 0.0) (42 . 0.0) (210 0.0 0.0 1.0)) 
If you use PEDIT to temporarily "close" that same polyline and again use the AutoLISP expression on it, the polyline might report: 
Command: (entget (car (entsel))) 
Select object: ((-1 . <Entity name: 40044e68>) (0 . "LWPOLYLINE") (330 . <Entity name: 40044cf8>) (5 . "65") (100 . "AcDbEntity") (67 . 0) (410 . "Model") (8 . "1") (100 . "AcDbPolyline") (90 . 4) (70 . 1) (43 . 0.0) (38 . 0.0) (39 . 0.0) (10 7.86986 10.6993) (40 . 0.0) (41 . 0.0) (42 . 0.0) (10 9.54074 7.86177) (40 . 0.0) (41 . 0.0) (42 . 0.0) (10 11.1782 10.232) (40 . 0.0) (41 . 0.0) (42 . 0.0) (10 9.57416 12.0347) (40 . 0.0) (41 . 0.0) (42 . 0.0) (210 0.0 0.0 1.0)) 
By examining the differences between these two reports, you can discover that Group Code 70 reports this "open/closed" state of a lightweight polyline and has a value of 1 when a lightweight polyline is "closed" and a value of 0 when it is "open." 
The following routine uses three Group Codes to filter the entire drawing for lightweight polylines (Group Code 0="LWPOLYLINE") on Layer "1" (Group Code 8="1") that are "closed" (Group Code 70=1) using the Select mode of a selection set object: 
Private Sub CommandButton3_Click()
   Dim entObj As AcadEntity
   Dim ssetObj As AcadSelectionSet
   Dim mode As Integer
   Dim grpCode(2) As Integer
   Dim dataVal(2) As Variant
   UserForm1.Hide
   Set ssetObj = ThisDrawing.SelectionSets.Add("SS01")
   mode = acSelectionSetAll
   grpCode(0) = 0
   dataVal(0) = "LWPOLYLINE"
   grpCode(1) = 70
   dataVal(1) = 1
   grpCode(2) = 8
   dataVal(2) = "1"
   ssetObj.Select mode, , , grpCode, dataVal
   ssetObj.Erase
   ssetObj.Delete
   UserForm1.Show
End Sub
As with any other aspect of using VBA to program routines for AutoCAD software, you'll soon begin to memorize commonly used Group Codes, and the filtering process gets easier and easier with each routine you write. Filtering entities also runs faster when the internal filtering parameters are used instead of nested loops that test individual entity properties. If this is your first exposure to Group Codes, spend some time examining the online DXF Reference for an overview of entity structures and purposes, and remember that applying the AutoLISP expression 
(entget (car (entsel))) 
on entities is a great way to report their Group Codes and Group Code states on-the-fly.
